Curvature and the geodesic equations

Copy/paste the following cell in a notebook and edit the data between the red lines. Then evaluate it to see the results.

Details on the code and how it works can be found at the end of the notebook.

In[ ]:=

```
Clear[coord, metric, inversemetric, affine, riemann, lriemann, uriemann,
  ricci, scalar, einstein, weyl, geodesic, R, G, τ, i, j, k, l, s];
Clear[r, θ, ϕ, t, χ, a, m, M];

(*--------------------------------------------------------------------------*)
(* This is what you need to set: *)
coord = {V, U, θ, ϕ};
n     = Length[coord];
metric = {
   {-(32 M^3/r[V, U]) e^(-r[V, U]/(2 M)), 0             , 0            ,                  0},
   {0              , (32 M^3/r[V, U]) e^(-r[V, U]/(2 M)), 0            ,                  0},
   {0              , 0                , r[V, U]^2     ,                  0},
   {0              , 0                , 0             , r[V, U]^2 Sin[θ]^2    }};
(*--------------------------------------------------------------------------*)

inversemetric = FullSimplify[Inverse[metric]];
Print["--------------------------------------------------------------------------"];
Print["The Manifold has dimension    n= ",
  n, "\nCoordinate system:           ", coord];
Print["--------------------------------------------------------------------------"];
Print["g_μν=", metric // MatrixForm];
Print["g^μν=", inversemetric // MatrixForm];
affine := affine = FullSimplify[Table[
      (1 / 2) * Sum[
        (*               g^is (∂_k g_sj+∂_j g_sk-∂_s g_jk)            *)
        (inversemetric[[i, s]]) *
         (D[metric[[s, j]], coord[[k]]] +
            D[metric[[s, k]], coord[[j]]] - D[metric[[j, k]], coord[[s]]]),
        {s, 1, n}],
      {i, 1, n}, {j, 1, n}, {k, 1, n}]];
(*The non zero Christoffel symbols are computed and selected below: *)
listaffine := Table[
   If[
```

```
     UnsameQ[affine⟦i, j, k⟧, 0],
     {Subscript[Superscript[Γ, i - 1], j - 1, k - 1], affine⟦i, j, k⟧}
      ],
    {i, 1, n}, {j, 1, n}, {k, 1, j}];
Print["-----------------------------------------------------------------------"];
Print["Christoffel Symbols:"];
Print[TableForm[
   Partition[DeleteCases[Flatten[listaffine], Null], 2], TableSpacing → {2, 2}]];
riemann := riemann = FullSimplify[Table[
      (* R^i_jkl=     ∂_kΓ^i_lj                          -           ∂_lΓ^i_kj*)
      D[   affine⟦i, l, j⟧,  coord⟦k⟧     ] - D[affine⟦i, k, j⟧, coord⟦l⟧    ] +
       (*           Γ^i_ks                 Γ^s_lj       -           Γ^i_ls                Γ^s_kj       *)
       Sum[affine⟦i, k, s⟧ affine⟦s, l, j⟧ -  affine⟦i, l, s⟧ affine⟦s, k, j⟧ ,
        {s, 1, n}],
      {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
listriemann := Table[
   If[
     UnsameQ[riemann⟦i, j, k, l⟧, 0],
     {Subscript[Superscript[R, i - 1], j - 1, k - 1, l - 1], riemann⟦i, j, k, l⟧}
    ],
    {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, k - 1}];
Print["-----------------------------------------------------------------------"];
Print["Riemann Tensor:"];
Print[TableForm[
   Partition[DeleteCases[Flatten[listriemann], Null], 2], TableSpacing → {2, 2}]];
lriemann := lriemann = FullSimplify[Table[
      Sum[metric⟦i, ii⟧ riemann⟦ii, j, k, l⟧, {ii, 1, n}],
      {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
listlriemann := Table[
   If[
     UnsameQ[lriemann⟦i, j, k, l⟧, 0],
     {Subscript[R, i - 1, j - 1, k - 1, l - 1], lriemann⟦i, j, k, l⟧}
    ], {i, 1, n}, {j, 1, i - 1}, {k, 1, n}, {l, 1, k - 1}];
Print["-----------------------------------------------------------------------"];
Print["Contravariant Riemann Tensor:"];
Print[TableForm[
   Partition[DeleteCases[Flatten[listlriemann], Null], 2], TableSpacing → {2, 2}]];
uriemann := uriemann = FullSimplify[Table[
      Sum[
       inversemetric⟦j, jj⟧ inversemetric⟦k, kk⟧ inversemetric⟦l, ll⟧
        riemann⟦i, jj, kk, ll⟧, {jj, 1, n}, {kk, 1, n}, {ll, 1, n}
      ],
```

```
      {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
listuriemann := Table[
    If[
     UnsameQ[uriemann[[i, j, k, l]], 0],
     {Superscript[Superscript[Superscript[Superscript[R, i - 1], j - 1], k - 1], l - 1],
      uriemann[[i, j, k, l]]}
    ] , {i, 1, n}, {j, 1, i - 1}, {k, 1, n}, {l, 1, k - 1}];
Print["------------------------------------------------------------------"];
Print["Covariant Riemann Tensor:"];
Print[TableForm[
    Partition[DeleteCases[Flatten[listuriemann], Null], 2], TableSpacing → {2, 2}]];
r2 = FullSimplify[Sum[lriemann[[i, j, k, l]] uriemann[[i, j, k, l]],
     {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
Print["------------------------------------------------------------------"];
Print["R^2= ", r2];
ricci := ricci = FullSimplify[Table[
      Sum[
       riemann[[i, j, i, l]],
       {i, 1, n}
      ], {j, 1, n}, {l, 1, n}] ];
listricci := Table[
    If[
     UnsameQ[ricci[[j, l]], 0],
     {Subscript[R, j - 1, l - 1], ricci[[j, l]]}
    ] , {j, 1, n}, {l, 1, j}];
Print["------------------------------------------------------------------"];
Print["Ricci Tensor:"];
Print[TableForm[
    Partition[DeleteCases[Flatten[listricci], Null], 2], TableSpacing → {2, 2}]];
scalar = FullSimplify[Sum[inversemetric[[i, j]] ricci[[i, j]], {i, 1, n}, {j, 1, n}] ];
Print["------------------------------------------------------------------"];
Print["Curvature Scalar:"];
Print["R= ", scalar];
einstein := einstein = FullSimplify[ricci - (1 / 2) scalar * metric];
listeinstein := Table[
    If[
     UnsameQ[einstein[[j, l]], 0],
     {Subscript[G, j - 1, l - 1], einstein[[j, l]]}
    ] , {j, 1, n}, {l, 1, j}];
Print["------------------------------------------------------------------"];
Print["Einstein Tensor:"];
Print[TableForm[
    Partition[DeleteCases[Flatten[listeinstein], Null], 2], TableSpacing → {2, 2}]];
```

```
weyl := weyl = FullSimplify[Table[
      If[n > 3,
       lriemann[[i, j, k, l]]
        -     1/(n - 2)     (metric[[i, k]] ricci[[l, j]] - metric[[i, l]] ricci[[k, j]] -
            metric[[j, k]] ricci[[l, i]] + metric[[j, l]] ricci[[k, i]])
        + 1/((n - 1) (n - 2)) (metric[[i, k]] metric[[l, j]] - metric[[i, l]] metric[[k, j]]) scalar
       (*else, if n≤ 3 return 0:*), 0],
      {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]];
listweyl := Table[
    If[
     UnsameQ[weyl[[i, j, k, l]], 0],
     {Subscript[C, i - 1, j - 1, k - 1, l - 1], weyl[[i, j, k, l]]}
    ], {i, 1, n}, {j, 1, i - 1}, {k, 1, n}, {l, 1, k - 1}];
Print["-----------------------------------------------------------------------"];
Print["Weyl Tensor:"];
Print[TableForm[
    Partition[DeleteCases[Flatten[listweyl], Null], 2], TableSpacing → {2, 2}]];
geodesic := geodesic =
    Simplify[Table[-Sum[affine[[i, j, k]] u[j] u[k], {j, 1, n}, {k, 1, n}], {i, 1, n}]];
subst = Table[u[i] → Subscript[coord[[i]], τ], {i, 1, n}];
nlistgeodesic :=
   Table[{Subscript[coord[[i]], ττ], "+", -geodesic[[i]] /. subst, "= 0"}, {i, 1, n}];
Print["-----------------------------------------------------------------------"];
Print["Geodesic Equations:"];
Print[TableForm[nlistgeodesic, TableSpacing → {2}]];
```

```
-----------------------------------------------------------------------
The Manifold has dimension   n= 4
Coordinate system:          {V, U, θ, ϕ}
-----------------------------------------------------------------------
```

$$g_{\mu\nu}=\begin{pmatrix} -\frac{32\, e^{-\frac{r[V,U]}{2M}} M^3}{r[V,U]} & 0 & 0 & 0 \\ 0 & \frac{32\, e^{-\frac{r[V,U]}{2M}} M^3}{r[V,U]} & 0 & 0 \\ 0 & 0 & r[V,U]^2 & 0 \\ 0 & 0 & 0 & r[V,U]^2 \operatorname{Sin}[\theta]^2 \end{pmatrix}$$

$$g^{\mu\nu} = \begin{pmatrix} -\frac{e^{\frac{r[V,U]}{2M}} r[V,U]}{32 M^3} & 0 & 0 & 0 \\ 0 & \frac{e^{\frac{r[V,U]}{2M}} r[V,U]}{32 M^3} & 0 & 0 \\ 0 & 0 & \frac{1}{r[V,U]^2} & 0 \\ 0 & 0 & 0 & \frac{\mathrm{Csc}[\theta]^2}{r[V,U]^2} \end{pmatrix}$$

Christoffel Symbols:

$\Gamma^{0}{}_{0,0}$ $-\frac{(2 M+r[V,U])\, r^{(1,0)}[V,U]}{4 M\, r[V,U]}$

$\Gamma^{0}{}_{1,0}$ $-\frac{(2 M+r[V,U])\, r^{(0,1)}[V,U]}{4 M\, r[V,U]}$

$\Gamma^{0}{}_{1,1}$ $-\frac{(2 M+r[V,U])\, r^{(1,0)}[V,U]}{4 M\, r[V,U]}$

$\Gamma^{0}{}_{2,2}$ $\frac{e^{\frac{r[V,U]}{2M}} r[V,U]^2\, r^{(1,0)}[V,U]}{32 M^3}$

$\Gamma^{0}{}_{3,3}$ $\frac{e^{\frac{r[V,U]}{2M}} r[V,U]^2\, \mathrm{Sin}[\theta]^2\, r^{(1,0)}[V,U]}{32 M^3}$

$\Gamma^{1}{}_{0,0}$ $-\frac{(2 M+r[V,U])\, r^{(0,1)}[V,U]}{4 M\, r[V,U]}$

$\Gamma^{1}{}_{1,0}$ $-\frac{(2 M+r[V,U])\, r^{(1,0)}[V,U]}{4 M\, r[V,U]}$

$\Gamma^{1}{}_{1,1}$ $-\frac{(2 M+r[V,U])\, r^{(0,1)}[V,U]}{4 M\, r[V,U]}$

$\Gamma^{1}{}_{2,2}$ $-\frac{e^{\frac{r[V,U]}{2M}} r[V,U]^2\, r^{(0,1)}[V,U]}{32 M^3}$

$\Gamma^{1}{}_{3,3}$ $-\frac{e^{\frac{r[V,U]}{2M}} r[V,U]^2\, \mathrm{Sin}[\theta]^2\, r^{(0,1)}[V,U]}{32 M^3}$

$\Gamma^{2}{}_{2,0}$ $\frac{r^{(1,0)}[V,U]}{r[V,U]}$

$\Gamma^{2}{}_{2,1}$ $\frac{r^{(0,1)}[V,U]}{r[V,U]}$

$\Gamma^{2}{}_{3,3}$ $-\mathrm{Cos}[\theta]\, \mathrm{Sin}[\theta]$

$\Gamma^{3}{}_{3,0}$ $\frac{r^{(1,0)}[V,U]}{r[V,U]}$

$\Gamma^{3}{}_{3,1}$ $\frac{r^{(0,1)}[V,U]}{r[V,U]}$

$\Gamma^{3}{}_{3,2}$ $\mathrm{Cot}[\theta]$

Riemann Tensor:

$$R^{0}{}_{1,1,0} \quad \frac{2\,M\,r^{(0,1)}[V,U]^2-2\,M\,r^{(1,0)}[V,U]^2+2\,M\,r[V,U]\,(-r^{(0,2)}[V,U]+r^{(2,0)}[V,U])+r[V,U]^2\,(-r^{(0,2)}[V,U]+r^{(2,0)}[V,U])}{4\,M\,r[V,U]^2}$$

$$R^{0}{}_{2,2,0} \quad -\frac{e^{\frac{r[V,U]}{2M}}\,r[V,U]\left((2\,M+r[V,U])\left(r^{(0,1)}[V,U]^2+r^{(1,0)}[V,U]^2\right)+4\,M\,r[V,U]\,r^{(2,0)}[V,U]\right)}{128\,M^4}$$

$$R^{0}{}_{2,2,1} \quad -\frac{e^{\frac{r[V,U]}{2M}}\,r[V,U]\,((2\,M+r[V,U])\,r^{(0,1)}[V,U]\,r^{(1,0)}[V,U]+2\,M\,r[V,U]\,r^{(1,1)}[V,U])}{64\,M^4}$$

$$R^{0}{}_{3,3,0} \quad -\frac{e^{\frac{r[V,U]}{2M}}\,r[V,U]\,\mathrm{Sin}[\theta]^2\left((2\,M+r[V,U])\left(r^{(0,1)}[V,U]^2+r^{(1,0)}[V,U]^2\right)+4\,M\,r[V,U]\,r^{(2,0)}[V,U]\right)}{128\,M^4}$$

$$R^{0}{}_{3,3,1} \quad -\frac{e^{\frac{r[V,U]}{2M}}\,r[V,U]\,\mathrm{Sin}[\theta]^2\,((2\,M+r[V,U])\,r^{(0,1)}[V,U]\,r^{(1,0)}[V,U]+2\,M\,r[V,U]\,r^{(1,1)}[V,U])}{64\,M^4}$$

$$R^{1}{}_{0,1,0} \quad \frac{2\,M\,r^{(0,1)}[V,U]^2-2\,M\,r^{(1,0)}[V,U]^2+2\,M\,r[V,U]\,(-r^{(0,2)}[V,U]+r^{(2,0)}[V,U])+r[V,U]^2\,(-r^{(0,2)}[V,U]+r^{(2,0)}[V,U])}{4\,M\,r[V,U]^2}$$

$$R^{1}{}_{2,2,0} \quad \frac{e^{\frac{r[V,U]}{2M}}\,r[V,U]\,((2\,M+r[V,U])\,r^{(0,1)}[V,U]\,r^{(1,0)}[V,U]+2\,M\,r[V,U]\,r^{(1,1)}[V,U])}{64\,M^4}$$

$$R^{1}{}_{2,2,1} \quad \frac{e^{\frac{r[V,U]}{2M}}\,r[V,U]\left((2\,M+r[V,U])\,r^{(0,1)}[V,U]^2+4\,M\,r[V,U]\,r^{(0,2)}[V,U]+(2\,M+r[V,U])\,r^{(1,0)}[V,U]^2\right)}{128\,M^4}$$

$$R^{1}{}_{3,3,0} \quad \frac{e^{\frac{r[V,U]}{2M}}\,r[V,U]\,\mathrm{Sin}[\theta]^2\,((2\,M+r[V,U])\,r^{(0,1)}[V,U]\,r^{(1,0)}[V,U]+2\,M\,r[V,U]\,r^{(1,1)}[V,U])}{64\,M^4}$$

$$R^{1}{}_{3,3,1} \quad \frac{e^{\frac{r[V,U]}{2M}}\,r[V,U]\,\mathrm{Sin}[\theta]^2\left((2\,M+r[V,U])\,r^{(0,1)}[V,U]^2+4\,M\,r[V,U]\,r^{(0,2)}[V,U]+(2\,M+r[V,U])\,r^{(1,0)}[V,U]^2\right)}{128\,M^4}$$

$$R^{2}{}_{0,2,0} \quad -\frac{(2\,M+r[V,U])\left(r^{(0,1)}[V,U]^2+r^{(1,0)}[V,U]^2\right)+4\,M\,r[V,U]\,r^{(2,0)}[V,U]}{4\,M\,r[V,U]^2}$$

$$R^{2}{}_{0,2,1} \quad -\frac{\frac{(2\,M+r[V,U])\,r^{(0,1)}[V,U]\,r^{(1,0)}[V,U]}{M}+2\,r[V,U]\,r^{(1,1)}[V,U]}{2\,r[V,U]^2}$$

$$R^{2}{}_{1,2,0} \quad -\frac{\frac{(2\,M+r[V,U])\,r^{(0,1)}[V,U]\,r^{(1,0)}[V,U]}{M}+2\,r[V,U]\,r^{(1,1)}[V,U]}{2\,r[V,U]^2}$$

$$R^{2}{}_{1,2,1} \quad -\frac{(2\,M+r[V,U])\,r^{(0,1)}[V,U]^2+4\,M\,r[V,U]\,r^{(0,2)}[V,U]+(2\,M+r[V,U])\,r^{(1,0)}[V,U]^2}{4\,M\,r[V,U]^2}$$

$$R^{2}{}_{3,3,2} \quad \frac{1}{32}\,\mathrm{Sin}[\theta]^2\left(-32+\frac{e^{\frac{r[V,U]}{2M}}\,r[V,U]\left(r^{(0,1)}[V,U]^2-r^{(1,0)}[V,U]^2\right)}{M^3}\right)$$

$$R^{3}{}_{0,3,0} \quad -\frac{(2\,M+r[V,U])\left(r^{(0,1)}[V,U]^2+r^{(1,0)}[V,U]^2\right)+4\,M\,r[V,U]\,r^{(2,0)}[V,U]}{4\,M\,r[V,U]^2}$$

$$R^{3}{}_{0,3,1} \quad -\frac{\frac{(2\,M+r[V,U])\,r^{(0,1)}[V,U]\,r^{(1,0)}[V,U]}{M}+2\,r[V,U]\,r^{(1,1)}[V,U]}{2\,r[V,U]^2}$$

$$R^{3}{}_{1,3,0} \quad -\frac{\frac{(2\,M+r[V,U])\,r^{(0,1)}[V,U]\,r^{(1,0)}[V,U]}{M}+2\,r[V,U]\,r^{(1,1)}[V,U]}{2\,r[V,U]^2}$$

$$R^{3}{}_{1,3,1} \quad -\frac{(2\,M+r[V,U])\,r^{(0,1)}[V,U]^2+4\,M\,r[V,U]\,r^{(0,2)}[V,U]+(2\,M+r[V,U])\,r^{(1,0)}[V,U]^2}{4\,M\,r[V,U]^2}$$

$$R^{3}{}_{2,3,2} \quad 1-\frac{e^{\frac{r[V,U]}{2M}}\,r[V,U]\left(r^{(0,1)}[V,U]^2-r^{(1,0)}[V,U]^2\right)}{32\,M^3}$$

Contravariant Riemann Tensor:

$$R_{1,0,1,0} \quad \frac{8\, e^{-\frac{r[V,U]}{2M}}\, M^2 \left(2\, M\, r^{(0,1)}[V,U]^2 - 2\, M\, r^{(1,0)}[V,U]^2 + 2\, M\, r[V,U] \left(-r^{(0,2)}[V,U] + r^{(2,0)}[V,U]\right) + r[V,U]^2 \left(-r^{(0,2)}[V,U] + r^{(2,0)}[V,U]\right)\right)}{r[V,U]^3}$$

$$R_{2,0,2,0} \quad -\frac{(2\, M + r[V,U]) \left(r^{(0,1)}[V,U]^2 + r^{(1,0)}[V,U]^2\right)}{4\, M} - r[V,\, U]\, r^{(2,0)}[V,\, U]$$

$$R_{2,0,2,1} \quad -\frac{(2\, M + r[V,U])\, r^{(0,1)}[V,U]\, r^{(1,0)}[V,U]}{2\, M} - r[V,\, U]\, r^{(1,1)}[V,\, U]$$

$$R_{2,1,2,0} \quad -\frac{(2\, M + r[V,U])\, r^{(0,1)}[V,U]\, r^{(1,0)}[V,U]}{2\, M} - r[V,\, U]\, r^{(1,1)}[V,\, U]$$

$$R_{2,1,2,1} \quad -\frac{(2\, M + r[V,U])\, r^{(0,1)}[V,U]^2 + 4\, M\, r[V,U]\, r^{(0,2)}[V,U] + (2\, M + r[V,U])\, r^{(1,0)}[V,U]^2}{4\, M}$$

$$R_{3,0,3,0} \quad -\frac{\mathrm{Sin}[\theta]^2 \left((2\, M + r[V,U]) \left(r^{(0,1)}[V,U]^2 + r^{(1,0)}[V,U]^2\right) + 4\, M\, r[V,U]\, r^{(2,0)}[V,U]\right)}{4\, M}$$

$$R_{3,0,3,1} \quad -\frac{1}{2}\, \mathrm{Sin}[\theta]^2 \left(\frac{(2\, M + r[V,U])\, r^{(0,1)}[V,U]\, r^{(1,0)}[V,U]}{M} + 2\, r[V,\, U]\, r^{(1,1)}[V,\, U]\right)$$

$$R_{3,1,3,0} \quad -\frac{1}{2}\, \mathrm{Sin}[\theta]^2 \left(\frac{(2\, M + r[V,U])\, r^{(0,1)}[V,U]\, r^{(1,0)}[V,U]}{M} + 2\, r[V,\, U]\, r^{(1,1)}[V,\, U]\right)$$

$$R_{3,1,3,1} \quad -\frac{\mathrm{Sin}[\theta]^2 \left((2\, M + r[V,U])\, r^{(0,1)}[V,U]^2 + 4\, M\, r[V,U]\, r^{(0,2)}[V,U] + (2\, M + r[V,U])\, r^{(1,0)}[V,U]^2\right)}{4\, M}$$

$$R_{3,2,3,2} \quad r[V,\, U]^2\, \mathrm{Sin}[\theta]^2 \left(1 - \frac{e^{\frac{r[V,U]}{2M}}\, r[V,U] \left(r^{(0,1)}[V,U]^2 - r^{(1,0)}[V,U]^2\right)}{32\, M^3}\right)$$

Covariant Riemann Tensor:

$$R^{1\,0\,1\,0} \quad \frac{e^{\frac{3\, r[V,U]}{2M}}\, r[V,U] \left(2\, M\, r^{(0,1)}[V,U]^2 - 2\, M\, r^{(1,0)}[V,U]^2 + 2\, M\, r[V,U] \left(-r^{(0,2)}[V,U] + r^{(2,0)}[V,U]\right) + r[V,U]^2 \left(-r^{(0,2)}[V,U] + r^{(2,0)}[V,U]\right)\right)}{131\,072\, M^{10}}$$

$$R^{2\,0\,2\,0} \quad -\frac{e^{\frac{r[V,U]}{M}} \left((2\, M + r[V,U]) \left(r^{(0,1)}[V,U]^2 + r^{(1,0)}[V,U]^2\right) + 4\, M\, r[V,U]\, r^{(2,0)}[V,U]\right)}{4096\, M^7\, r[V,U]^2}$$

$$R^{2\,0\,2\,1} \quad \frac{e^{\frac{r[V,U]}{M}} \left((2\, M + r[V,U])\, r^{(0,1)}[V,U]\, r^{(1,0)}[V,U] + 2\, M\, r[V,U]\, r^{(1,1)}[V,U]\right)}{2048\, M^7\, r[V,U]^2}$$

$$R^{2\,1\,2\,0} \quad \frac{e^{\frac{r[V,U]}{M}} \left((2\, M + r[V,U])\, r^{(0,1)}[V,U]\, r^{(1,0)}[V,U] + 2\, M\, r[V,U]\, r^{(1,1)}[V,U]\right)}{2048\, M^7\, r[V,U]^2}$$

$$R^{2\,1\,2\,1} \quad -\frac{e^{\frac{r[V,U]}{M}} \left((2\, M + r[V,U])\, r^{(0,1)}[V,U]^2 + 4\, M\, r[V,U]\, r^{(0,2)}[V,U] + (2\, M + r[V,U])\, r^{(1,0)}[V,U]^2\right)}{4096\, M^7\, r[V,U]^2}$$

$$R^{3\,0\,3\,0} \quad -\frac{e^{\frac{r[V,U]}{M}}\, \mathrm{Csc}[\theta]^2 \left((2\, M + r[V,U]) \left(r^{(0,1)}[V,U]^2 + r^{(1,0)}[V,U]^2\right) + 4\, M\, r[V,U]\, r^{(2,0)}[V,U]\right)}{4096\, M^7\, r[V,U]^2}$$

$$R^{3\,0\,3\,1} \quad \frac{e^{\frac{r[V,U]}{M}}\, \mathrm{Csc}[\theta]^2 \left((2\, M + r[V,U])\, r^{(0,1)}[V,U]\, r^{(1,0)}[V,U] + 2\, M\, r[V,U]\, r^{(1,1)}[V,U]\right)}{2048\, M^7\, r[V,U]^2}$$

$$R^{3\,1\,3\,0} \quad \frac{e^{\frac{r[V,U]}{M}}\, \mathrm{Csc}[\theta]^2 \left((2\, M + r[V,U])\, r^{(0,1)}[V,U]\, r^{(1,0)}[V,U] + 2\, M\, r[V,U]\, r^{(1,1)}[V,U]\right)}{2048\, M^7\, r[V,U]^2}$$

$$R^{3\,1\,3\,1} \quad -\frac{e^{\frac{r[V,U]}{M}}\, \mathrm{Csc}[\theta]^2 \left((2\, M + r[V,U])\, r^{(0,1)}[V,U]^2 + 4\, M\, r[V,U]\, r^{(0,2)}[V,U] + (2\, M + r[V,U])\, r^{(1,0)}[V,U]^2\right)}{4096\, M^7\, r[V,U]^2}$$

$$R^{3\,2\,3\,2} \quad \frac{\mathrm{Csc}[\theta]^2 \left(1 - \frac{e^{\frac{r[V,U]}{2M}}\, r[V,U] \left(r^{(0,1)}[V,U]^2 - r^{(1,0)}[V,U]^2\right)}{32\, M^3}\right)}{r[V,U]^6}$$

$$R^2= \frac{4}{r[V,U]^4}+\frac{e^{\frac{r[V,U]}{2M}}\left(-r^{(0,1)}[V,U]^2+r^{(1,0)}[V,U]^2\right)}{4M^3\,r[V,U]^3}+\frac{1}{4096\,M^8\,r[V,U]^2}$$

$$e^{\frac{r[V,U]}{M}}\Big(4\left(9M^2+4M\,r[V,U]+r[V,U]^2\right)r^{(0,1)}[V,U]^4+36M^2\,r^{(1,0)}[V,U]^4-64M\,r[V,U]\,(2M+r[V,U])\,r^{(0,1)}[V,U]$$
$$r^{(1,0)}[V,U]\,r^{(1,1)}[V,U]+4M\,r[V,U]^3\left(r^{(0,2)}[V,U]-r^{(2,0)}[V,U]\right)^2+r[V,U]^4\left(r^{(0,2)}[V,U]-r^{(2,0)}[V,U]\right)^2+$$
$$8M\,r[V,U]\,r^{(1,0)}[V,U]^2\left(5M\,r^{(0,2)}[V,U]+2\,r^{(1,0)}[V,U]^2+3M\,r^{(2,0)}[V,U]\right)+$$
$$4\,r^{(0,1)}[V,U]^2\Big(-18M^2\,r^{(1,0)}[V,U]^2+2M\,r[V,U]\left(3M\,r^{(0,2)}[V,U]-4\,r^{(1,0)}[V,U]^2+5M\,r^{(2,0)}[V,U]\right)+$$
$$r[V,U]^2\left(3M\,r^{(0,2)}[V,U]-2\,r^{(1,0)}[V,U]^2+5M\,r^{(2,0)}[V,U]\right)\Big)+$$
$$4\,r[V,U]^2\Big(9M^2\,r^{(0,2)}[V,U]^2+r^{(1,0)}[V,U]^4+3M\,r^{(1,0)}[V,U]^2\,r^{(2,0)}[V,U]+$$
$$M\,r^{(0,2)}[V,U]\left(5\,r^{(1,0)}[V,U]^2-2M\,r^{(2,0)}[V,U]\right)+M^2\left(-16\,r^{(1,1)}[V,U]^2+9\,r^{(2,0)}[V,U]^2\right)\Big)\Big)$$

Ricci Tensor:

$$R_{0,0}\quad \frac{-2\,(M+r[V,U])\,r^{(0,1)}[V,U]^2-6M\,r^{(1,0)}[V,U]^2+r[V,U]^2\left(-r^{(0,2)}[V,U]+r^{(2,0)}[V,U]\right)-2\,r[V,U]\left(M\,r^{(0,2)}[V,U]+r^{(1,0)}[V,U]^2+3M\,r^{(2,0)}[V,U]\right)}{4M\,r[V,U]^2}$$

$$R_{1,0}\quad -\frac{\frac{(2M+r[V,U])\,r^{(0,1)}[V,U]\,r^{(1,0)}[V,U]}{M}+2\,r[V,U]\,r^{(1,1)}[V,U]}{r[V,U]^2}$$

$$R_{1,1}\quad \frac{-2\,(3M+r[V,U])\,r^{(0,1)}[V,U]^2-2M\,r^{(1,0)}[V,U]^2+r[V,U]^2\left(r^{(0,2)}[V,U]-r^{(2,0)}[V,U]\right)-2\,r[V,U]\left(3M\,r^{(0,2)}[V,U]+r^{(1,0)}[V,U]^2+M\,r^{(2,0)}[V,U]\right)}{4M\,r[V,U]^2}$$

$$R_{2,2}\quad 1-\frac{e^{\frac{r[V,U]}{2M}}\,r[V,U]\left(r^{(0,1)}[V,U]^2-r^{(1,0)}[V,U]^2+r[V,U]\left(r^{(0,2)}[V,U]-r^{(2,0)}[V,U]\right)\right)}{32M^3}$$

$$R_{3,3}\quad \mathrm{Sin}[\theta]^2-\frac{e^{\frac{r[V,U]}{2M}}\,r[V,U]\,\mathrm{Sin}[\theta]^2\left(r^{(0,1)}[V,U]^2-r^{(1,0)}[V,U]^2+r[V,U]\left(r^{(0,2)}[V,U]-r^{(2,0)}[V,U]\right)\right)}{32M^3}$$

Curvature Scalar:

$$R=\frac{128+\frac{e^{\frac{r[V,U]}{2M}}\,r[V,U]\left(-6M\,r^{(0,1)}[V,U]^2+6M\,r^{(1,0)}[V,U]^2+r[V,U]^2\left(r^{(0,2)}[V,U]-r^{(2,0)}[V,U]\right)+6M\,r[V,U]\left(-r^{(0,2)}[V,U]+r^{(2,0)}[V,U]\right)\right)}{M^4}}{64\,r[V,U]^2}$$

Einstein Tensor:

$$G_{0,0}\quad -\frac{-64\,e^{-\frac{r[V,U]}{2M}}M^3+\frac{r[V,U]\left((4M+r[V,U])\,r^{(0,1)}[V,U]^2+r[V,U]\left(4M\,r^{(0,2)}[V,U]+r^{(1,0)}[V,U]^2\right)\right)}{M}}{2\,r[V,U]^3}$$

$$G_{1,0}\quad -\frac{\frac{(2M+r[V,U])\,r^{(0,1)}[V,U]\,r^{(1,0)}[V,U]}{M}+2\,r[V,U]\,r^{(1,1)}[V,U]}{r[V,U]^2}$$

$$G_{1,1}\quad -\frac{64\,e^{-\frac{r[V,U]}{2M}}M^3+4\,r[V,U]\,r^{(1,0)}[V,U]^2+\frac{r[V,U]^2\left(r^{(0,1)}[V,U]^2+r^{(1,0)}[V,U]^2+4M\,r^{(2,0)}[V,U]\right)}{M}}{2\,r[V,U]^3}$$

$$G_{2,2}\quad \frac{e^{\frac{r[V,U]}{2M}}\,r[V,U]\left(2M\,r^{(0,1)}[V,U]^2-2M\,r^{(1,0)}[V,U]^2+2M\,r[V,U]\left(r^{(0,2)}[V,U]-r^{(2,0)}[V,U]\right)+r[V,U]^2\left(-r^{(0,2)}[V,U]+r^{(2,0)}[V,U]\right)\right)}{128M^4}$$

$$G_{3,3}\quad \frac{e^{\frac{r[V,U]}{2M}}\,r[V,U]\,\mathrm{Sin}[\theta]^2\left(2M\,r^{(0,1)}[V,U]^2-2M\,r^{(1,0)}[V,U]^2+2M\,r[V,U]\left(r^{(0,2)}[V,U]-r^{(2,0)}[V,U]\right)+r[V,U]^2\left(-r^{(0,2)}[V,U]+r^{(2,0)}[V,U]\right)\right)}{128M^4}$$

Weyl Tensor:

$C_{1,0,1,0}$ $$\frac{e^{-\frac{r[V,U]}{M}}\left(-1024\,M^6-8\,e^{\frac{r[V,U]}{2M}}\,M^2\,r[V,U]\left(-6\,M\,r^{(0,1)}[V,U]^2+6\,M\,r[V,U]\,r^{(0,2)}[V,U]+r[V,U]^2\,r^{(0,2)}[V,U]+6\,M\,r^{(1,0)}[V,U]^2-r[V,U]\,(6\,M+r[V,U])\,r^{(2,0)}[V,U]\right)\right)}{3\,r[V,U]^4}$$

$C_{2,0,2,0}$ $$\frac{1}{24}\left(\frac{128\,e^{-\frac{r[V,U]}{2M}}\,M^3}{r[V,U]}-6\,r^{(0,1)}[V,U]^2+6\,r^{(1,0)}[V,U]^2+\frac{r[V,U]\,(6\,M+r[V,U])\,(r^{(0,2)}[V,U]-r^{(2,0)}[V,U])}{M}\right)$$

$C_{2,1,2,1}$ $$\frac{1}{24}\left(-\frac{128\,e^{-\frac{r[V,U]}{2M}}\,M^3}{r[V,U]}+6\left(r^{(0,1)}[V,U]^2-r^{(1,0)}[V,U]^2\right)-\frac{r[V,U]\,(6\,M+r[V,U])\,(r^{(0,2)}[V,U]-r^{(2,0)}[V,U])}{M}\right)$$

$C_{3,0,3,0}$ $$\frac{1}{24}\,\mathrm{Sin}[\theta]^2\left(\frac{128\,e^{-\frac{r[V,U]}{2M}}\,M^3}{r[V,U]}-6\,r^{(0,1)}[V,U]^2+6\,r^{(1,0)}[V,U]^2+\frac{r[V,U]\,(6\,M+r[V,U])\,(r^{(0,2)}[V,U]-r^{(2,0)}[V,U])}{M}\right)$$

$C_{3,1,3,1}$ $$\frac{1}{24}\,\mathrm{Sin}[\theta]^2\left(-\frac{128\,e^{-\frac{r[V,U]}{2M}}\,M^3}{r[V,U]}+6\left(r^{(0,1)}[V,U]^2-r^{(1,0)}[V,U]^2\right)-\frac{r[V,U]\,(6\,M+r[V,U])\,(r^{(0,2)}[V,U]-r^{(2,0)}[V,U])}{M}\right)$$

$C_{3,2,3,2}$ $$\frac{r[V,U]^2\,\mathrm{Sin}[\theta]^2\left(128\,M^4+e^{\frac{r[V,U]}{2M}}\,r[V,U]\left(-6\,M\,r^{(0,1)}[V,U]^2+6\,M\,r[V,U]\,r^{(0,2)}[V,U]+r[V,U]^2\,r^{(0,2)}[V,U]+6\,M\,r^{(1,0)}[V,U]^2-r[V,U]\,(6\,M+r[V,U])\,r^{(2,0)}[V,U]\right)\right)}{384\,M^4}$$

Geodesic Equations:

$$V_{\tau\tau} + -\frac{16\,M^2\,(2\,M+r[V,U])\,U_\tau\,V_\tau\,r^{(0,1)}[V,U]+\left(16\,M^3\,(U_\tau^2+V_\tau^2)+8\,M^2\,r[V,U]\,(U_\tau^2+V_\tau^2)-e^{\frac{r[V,U]}{2M}}\,r[V,U]^3\,(\theta_\tau^2+\mathrm{Sin}[\theta]^2\,\phi_\tau^2)\right)r^{(1,0)}[V,U]}{32\,M^3\,r[V,U]} = 0$$

$$U_{\tau\tau} + -\frac{\left(16\,M^3\,(U_\tau^2+V_\tau^2)+8\,M^2\,r[V,U]\,(U_\tau^2+V_\tau^2)+e^{\frac{r[V,U]}{2M}}\,r[V,U]^3\,(\theta_\tau^2+\mathrm{Sin}[\theta]^2\,\phi_\tau^2)\right)r^{(0,1)}[V,U]+16\,M^2\,(2\,M+r[V,U])\,U_\tau\,V_\tau\,r^{(1,0)}[V,U]}{32\,M^3\,r[V,U]} = 0$$

$$\theta_{\tau\tau} + -\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\,\phi_\tau^2+\frac{2\,\theta_\tau\,(U_\tau\,r^{(0,1)}[V,U]+V_\tau\,r^{(1,0)}[V,U])}{r[V,U]} = 0$$

$$\phi_{\tau\tau} + \frac{2\,\phi_\tau\,(\mathrm{Cot}[\theta]\,r[V,U]\,\theta_\tau+U_\tau\,r^{(0,1)}[V,U]+V_\tau\,r^{(1,0)}[V,U])}{r[V,U]} = 0$$

Postscript

The code in this notebook is based on James Hartle’s Mathematica programs written as companions to his book “Gravity: An Introduction to Einstein’s General Relativity”, available from the Book’s site: http://web.physics.ucsb.edu/~gravitybook/mathematica.html
The original program was written by *Leonard Parker, University of Wisconsin, Milwaukee* (see the Acknowledgement section in the end of this notebook). The code in this notebook is a modification of the original program, adapted for the needs of the course “General Relativity and Cosmology” offered at the National Technical University of Athens, by Konstantinos Anagnostopoulos (http://physics.ntua.gr/konstant).The site of the course for the Spring 2023 can be found at *http://physics.ntua.gr/konstant/GR* .

This is the *Mathematica* notebook *Curvature and the Einstein Equation* available from the book website. From a given metric $g_{\alpha\beta}$, it computes the components of the following: the inverse metric, $g^{\lambda\sigma}$, the Christoffel symbols or affine connection,

$$\Gamma^\lambda{}_{\mu\nu} = \frac{1}{2}\,g^{\lambda\sigma}(\partial_\mu\,g_{\sigma\nu}+\partial_\nu\,g_{\sigma\mu}-\partial_\sigma\,g_{\mu\nu}),$$

(∂_α stands for the partial derivative $\partial/\partial x^\alpha$), the Riemann tensor,

$R^\lambda{}_{\mu\nu\sigma} = \partial_\nu \Gamma^\lambda{}_{\mu\sigma} - \partial_\sigma \Gamma^\lambda{}_{\mu\nu} + \Gamma^\lambda{}_{\eta\nu} \Gamma^\eta{}_{\mu\sigma} - \Gamma^\lambda{}_{\eta\sigma} \Gamma^\eta{}_{\mu\nu}$,

the Ricci tensor

$R_{\mu\nu} = R^\lambda{}_{\mu\lambda\nu}$,

the scalar curvature,

$R = g^{\mu\nu} R_{\mu\nu}$,

and the Einstein tensor,

$G_{\mu\nu} = R_{\mu\nu} - \frac{1}{2} g_{\mu\nu} R$.

You must input the covariant components of the metric tensor $g_{\mu\nu}$ by editing the relevant input line in this *Mathematica* notebook. You may also wish to change the names of the coordinates. Only the nonzero components of the above quantities are displayed as the output. All the components computed are in the *coordinate basis* in which the metric was specified.